
Strachey's functional pearl, forty years on

Mike Spivey, July 2006

In a 1966 paper, Christopher Strachey gave a number of examples of programs written in the programming language CPL using a functional style, and among them is a remarkable definition of a Cartesian product function `cprod` on lists of lists that uses (the function we now know as) `foldr` instead of recursion. Strachey presents this definition without comment or derivation. This note gives a derivation from a specification in the form of a list comprehension.

The function `cprod` takes a list of lists and computes its Cartesian product:

```
cprod [[1, 2, 3], [4], [5, 6]]
= [[1, 4, 5], [1, 4, 6], [2, 4, 5], [2, 4, 6], [3, 4, 5], [3, 4, 6]]
```

We can specify this function using recursion and list comprehension:

```
cprod [] = [[]]
cprod (xs : zss) = [ x : ys | x ← xs, ys ← cprod zss ]
```

Now what we want is a version of `cprod` that uses no recursion, but is written instead in terms of `foldr` and elementary operations on lists.

The main principle to which we shall appeal is this: that $p = \text{foldr } f \ a$ is the unique solution for p of the equations,

$$\begin{aligned} p [] &= a \\ p (x : xs) &= f \ x \ (p \ xs) \end{aligned}$$

For example, if we take $p \ xs = \text{concat } (\text{map } f_0 \ xs)$, then $p [] = []$ and $p (x : xs) = f_0 \ x \ ++ \ p \ xs$, so we can deduce that $p = \text{foldr } f \ []$, where $f \ x \ ys = f_0 \ x \ ++ \ ys$.

It is plain that `cprod` itself can be written as a `foldr`, because `cprod (xs : zss)` depends on `zss` only through `cprod zss`. Thus:

$$\text{cprod} = \text{foldr } f \ [[]]$$

where $f \ xs \ yss = [x : ys \mid x \leftarrow xs, ys \leftarrow yss]$. It remains to express this function also in terms of `foldr`. The meaning of the two-generator list comprehension for f can be expressed using nested comprehensions and `concat`:

$$\begin{aligned} f \ xs \ yss &= \text{concat } [[x : ys \mid ys \leftarrow yss] \mid x \leftarrow xs] \\ &= \text{concat } (\text{map } (g_0 \ yss) \ xs) \end{aligned}$$

2 Strachey's functional pearl, or ty years on

where $g_0 \text{ yss } x = [x : ys \mid ys \leftarrow \text{yss}] = \text{map } (x:) \text{ yss}$. (The nesting of the ys comprehension inside the x comprehension reflects the fact that ys “varies faster than” x .) Now we can use the result about *concat* and *map* mentioned above: this gives

$$f \text{ xs yss} = \text{foldr } (g \text{ yss}) [] \text{ xs}$$

where $g \text{ yss } x \text{ zss} = g_0 \text{ yss } x ++ \text{zss} = \text{map } (x:) \text{ yss} ++ \text{zss}$.

In its turn, we look for a definition of g in terms of *foldr*. Clearly,

$$\begin{aligned} g [] x \text{ zss} &= \text{zss}, \\ g (ys : yss) x \text{ zss} &= \text{map } (x:) (ys : yss) ++ \text{zss} \\ &= (x : ys) : (\text{map } (x:) \text{ yss} ++ \text{zss}) = (x : ys) : (g \text{ yss } x \text{ zss}), \end{aligned}$$

so $g \text{ yss } x \text{ zss} = \text{foldr } (h \text{ x}) \text{ zss } \text{yss}$ where $h \text{ x } ys \text{ uss} = (x : ys) : \text{uss}$.

Putting it all together, we obtain

$$\begin{aligned} cprod &= \text{foldr } f [[]] \\ \mathbf{where} \\ f \text{ xs yss} &= \text{foldr } (g \text{ yss}) [] \text{ xs} \\ g \text{ yss } x \text{ zss} &= \text{foldr } (h \text{ x}) \text{ zss } \text{yss} \\ h \text{ x } ys \text{ uss} &= (x : ys) : \text{uss} \end{aligned}$$

To smooth the derivation I have made the free variables of each auxiliary function body explicit as extra arguments, but they can be eliminated in favour of nested **where** clauses to obtain essentially the program that Strachey wrote:

$$\begin{aligned} cprod &= \text{foldr } f [[]] \\ \mathbf{where} \quad f \text{ xs yss} &= \text{foldr } g [] \text{ xs} \\ \mathbf{where} \quad g \text{ x zss} &= \text{foldr } h \text{ zss } \text{yss} \\ \mathbf{where} \quad h \text{ ys uss} &= (x : ys) : \text{uss} \end{aligned}$$

Olivier Danvy suggested that Strachey's program could be seen as the earliest functional pearl, and I had already been using it as the first exercise in my course on programming languages at Oxford. This note is an expanded version of the model solution of that problem. Strachey's program had already been recalled to attention in recent times in Mike Gordon's reminiscence in a special issue of the journal Higher-Order and Symbolic Computation published in 2000 to commemorate the 25th anniversary of Strachey's death.